



★ DHT11-Humidity-TempSensor

Page ▾

Discussion

History


Notify Me

PROTECTED

Digital output temperature and humidity sensor DHT11

(Available here: http://arduino-direct.com/sunshop/index.php?l=product_detail&p=162)

See photo at right: Connections are (V)oltage, (S)ignal, (G)round



- Full range of calibration, in-line digital output;
- Humidity measuring range: 20% ~ 90% RH (0-50 °C temperature compensation)
- Temperature measuring range: 0 ~ +50 °C ;
- Humidity measuring accuracy: 5.0% RH
- Temperature measurement accuracy: 2.0 °C
- Response time: <5S ;
- Low power consumption

Features

.. single wire digital interface (the most simple system integration, ultra-low prices)

.. ultra-small size (12X15.5X5.5 mm)

.. high reliability

.. optimized long-term stability

Test Software Sketch for DHT11 (Credits: Rob Tillaart)

Search

HOME

YourDuinoShop

ELECTRONIC BRICKS:

- STARTER SET
- Sensor Shield
- Bricks List

ARDUINO INFO:

- Starting Software
- ARDUINO POWER!

TUTORIALS

- Ultrasonic Sensors
- Stepper Motors
- RealTimeClock
- PWM Frequencies

BJK
2DO

```

/* YourDuino.com Example Software Sketch
   DHT11 Humidity and Temperature Sensor test
   Credits: Rob Tillaart
   http://arduino-direct.com/sunshop/index.php?l=product_detail&p=162
   terry@yourduino.com */

/*-----( Import needed libraries )-----*/
#include <dht11.h>

/*-----( Declare objects )-----*/
dht11 DHT11;

/*-----( Declare Constants, Pin Numbers )-----*/
#define DHT11PIN 2

void setup() /*-----( SETUP: RUNS ONCE )-----*/
{
  Serial.begin(9600);
  Serial.println("DHT11 TEST PROGRAM ");
  Serial.print("LIBRARY VERSION: ");
  Serial.println(DHT11LIB_VERSION);
  Serial.println();
}/*--(end setup )---*/

void loop() /*-----( LOOP: RUNS CONSTANTLY )-----*/
{
  Serial.println("\n");

  int chk = DHT11.read(DHT11PIN);

  Serial.print("Read sensor: ");
  switch (chk)
  {
    case 0: Serial.println("OK"); break;
    case -1: Serial.println("Checksum error"); break;
    case -2: Serial.println("Time out error"); break;
    default: Serial.println("Unknown error"); break;
  }

  Serial.print("Humidity (%): ");
  Serial.println((float)DHT11.humidity, 2);

  Serial.print("Temperature (oC): ");
  Serial.println((float)DHT11.temperature, 2);

  Serial.print("Temperature (oF): ");
  Serial.println(Fahrenheit(DHT11.temperature), 2);

  Serial.print("Temperature (K): ");
  Serial.println(Kelvin(DHT11.temperature), 2);

  Serial.print("Dew Point (oC): ");
  Serial.println(dewPoint(DHT11.temperature, DHT11.humidity));

  Serial.print("Dew PointFast (oC): ");
  Serial.println(dewPointFast(DHT11.temperature, DHT11.humidity));

  delay(2000);
}/* --(end main loop )-- */

/*-----( Declare User-written Functions )-----*/
//
//Celsius to Fahrenheit conversion
double Fahrenheit(double celsius)
{
  return 1.8 * celsius + 32;
}

//Celsius to Kelvin conversion
double Kelvin(double celsius)
{
  return celsius + 273.15;
}

```

```

// dewPoint function NOAA
// reference: http://wahiduddin.net/calc/density_algorithms.htm
double dewPoint(double celsius, double humidity)
{
    double A0= 373.15/(273.15 + celsius);
    double SUM = -7.90298 * (A0-1);
    SUM += 5.02808 * log10(A0);
    SUM += -1.3816e-7 * (pow(10, (11.344*(1-1/A0)))-1) ;
    SUM += 8.1328e-3 * (pow(10, (-3.49149*(A0-1)))-1) ;
    SUM += log10(1013.246);
    double VP = pow(10, SUM-3) * humidity;
    double T = log(VP/0.61078); // temp var
    return (241.88 * T) / (17.558-T);
}

// delta max = 0.6544 wrt dewPoint()
// 5x faster than dewPoint()
// reference: http://en.wikipedia.org/wiki/Dew_point
double dewPointFast(double celsius, double humidity)
{
    double a = 17.271;
    double b = 237.7;
    double temp = (a * celsius) / (b + celsius) + log(humidity/100);
    double Td = (b * temp) / (a - temp);
    return Td;
}

/* ( THE END ) */

```

The output from this program to the Serial Monitor will be something like this:

DHT11 TEST PROGRAM

LIBRARY VERSION: 0.3.2

Read sensor: OK

Humidity (%): 56.00

Temperature (oC): 23.00

Temperature (oF): 73.40

Temperature (K): 296.15

Dew Point (oC): 13.77

Dew PointFast (oC): 13.74

DETAILS: Before you can use an "Arduino Software Library", you must make it available by copying it into the correct "libraries" folder. From the playground pages mentioned above:

Notes

To use the library, unzip the file below into your SKETCHBOOKPATH\libraries.

Here is a .ZIP file available with these files for easier installation:

[DHT11.zip](#)



